
'*******************************************************************

'*  Name    : Automatic Water Controler Circuit (using fuzzy logic)*

'*  Author  : Roy H. Guerra Jr.                                    *

'*  Notice  : Copyright (c) 2009 .                                 *

'*          : All Rights Reserved                                  *

'*  Date    : 3/10/2009                                            *

'*  Version : 1.0                                                  *

'*  Notes   : Use PIC12F675                                        *

'*          :                                                      *  

'*******************************************************************

'

' 

' Hardware Interfaces:

' ====================

' 1. CDS Photocell input on GP2 (input)

'

' 2. Water Valve Solenoid fed from FET Transistor on GP0 (output)

'

' 3. Moisture sense probes on GP4 and GP5 (GP4 input and GP5 output)        

' 4. MCLR Pin (GP3) is tied to 0.1uF capacitor to reset when power

'    is first turned on.    

'

' 5. A WatchDog Timer Fault illuminates the "Fault LED"

'

'

' Device Programming Modes (Fuses) 

' ================================

' Internal Oscillator (RC) 

' Enable Watch Dog Timer

' Disable Power-up timer

' Disable Brown-out detect

' Reset on GP3 Enabled

' GP4 & GP5 as I/O

' EE Program Disabled

'  

'

' Software Declarations

' =====================

'

' Define A/D Parameters

' ---------------------

DEFINE  ADC_BITS    10  ' Set number of bits in result

DEFINE  ADC_CLOCK   3   ' Set clock source (3=rc)

DEFINE  ADC_SAMPLEUS 50 ' Set sampling time in uS

'

' Special Register & Interrupt Declarations

' -----------------------------------------   
OPTION_REG = %10000000  ' Disable PORTB pull-ups 
INTCON = %00000000  ' Disable Interrupts
WPU = %00000000     ' Disable Resistor Pullup's
VRCON.7 = 0         ' Disable Voltage Reference
TRISIO = %00011100  ' GP0, GP1, GP5 are outputs, GP2, GP3 & GP4 are inputs
ADCON0.7 = 1        ' Right justify 10 bit A/D conversion
ANSEL = %01111100   ' GP2&GP4 set to Anolog Inputs, use Internal RC Oscillator 
CMCON = 7       ' Analog comparators off

'

'

' Declare Program Variables

' -------------------------
Adval1 VAR WORD     ' Temp RAM storage for GP2 A/D conversion value
Adval2 VAR WORD     ' Temp RAM storage for GP4 A/D conversion value
I VAR BYTE      ' Temp Storage variable for loop counting 
J VAR BYTE      ' Temp Storage variable for loop counting 

'

'

' Declare Program Modifiers

' -------------------------    
Water VAR GPIO.0    ' Name for GP0 Port
Fault VAR GPIO.1    ' Name for GP1 Port

'

'   

' Declare Initial Power On States

' -------------------------------
Water = 0       ' Set Initial Power Up States to Off
Fault = 0

I = 0

J = 0

'

'

' Main Program Start

' ==================

'
Loop:

    Water = 0   ' Ensures water not on
    Fault = 0   ' Ensures fault LED not on
    GOSUB AtoD  ' Go to A/D subroutine
    GOSUB Transmit  ' Go to Transmit subroutine
    GOSUB Recieve   ' Go to Recieve subroutine
    GOSUB Error ' Go to Error subroutine
    

GOTO Loop               ' Back to Program Beginning  

END

'

'

' A/D Subroutine

'---------------

' Notes - Scale equation is as follows:

' A/D resolution is 2^10 = 1024

' 5 Volts = 1024 (using A/D reference of VSS), Digital = Vin * (1024/ 5)

' Scale using a multiplier value of 500 for best accuracy & adlusting Digital

' scaling (0-500) 

' Scale equation = Vin * [((1024/5) * 500) / 1024] = Vin * 100

' Adval1 is Dark = 1.8 volts "Vin" so; 1.8 * 100 = 180.  Ensuring a 20% margin,

' gives 216 

' Adval2 is low soil conductivity = 0.5 volt "Vin" so 0.5 * 100 = 50 no margin 

' required, adjust Pot.
AtoD:   ADCIN 2, Adval1     ' Read channel 2 to adval1
    PAUSE 100       ' Delay 100mS for conversion time
    Adval1 = (Adval1 */ 500)>>2 ' equates to: (Adval1 * 500)/1024 
    PAUSE 10        ' Small delay

    ADCIN 3, Adval2     ' Read channel 3 to adval2
    PAUSE 100       ' Delay 100mS for conversion time
    Adval2 = (Adval2 */ 500)>>2 ' equates to: (Adval2 * 500)/1024 
    PAUSE 10        ' Small delay
    RETURN          ' Return to main program

'

'

'Transmit Subroutine

'-------------------

' Pulse of 10mS on and 1 mS off with a repitition rate of "10"
Transmit:

    FOR I = 0 TO 9  'perform pulse transmission 10 times 
    PULSOUT GPIO.5, 1000    ' (100 * 10uS = 10 mS)  
    PAUSE 1     ' Off time
    NEXT I      ' Next pulse
    I = 0       ' Clear counter
    RETURN      ' Return to main program

'

'

'Recieve Subroutine

'-------------------

' Uses fuzzy logic to determine when to turn on water soleniod and duration

' Fuzzy Logic Rules:

' 1) If getting dark and soil very dry then water very long

' 2) If getting dark and soil dry then water medium

' 3) If getting dark and soil semi dry then water low

' 4) If light or soil wet then do not water
Recieve:

    IF (Adval1 < 216) AND (Adval2 < 50) THEN    ' Rule #1
        FOR J = 0 TO 59     ' (60 * 1 minute = 1 Hour) 
        Water = 1               ' Turn on water soleniod 
        PAUSE 60000 ' 1 minute time delay 
        NEXT J

    ENDIF
    J = 0       ' Clear counter
          

    IF (Adval1 < 216) AND (Adval2 > 50 AND Adval2 < 150) THEN   ' Rule #2
        FOR J = 0 TO 29     ' (30 * 1 minute = 0.5 Hour) 
        Water = 1               ' Turn on water soleniod 
        PAUSE 60000 ' 1 minute time delay 
        NEXT J

    ENDIF
    J = 0       ' Clear counter
        

    IF (Adval1 < 216) AND (Adval2 > 150 AND Adval2 < 250 ) THEN ' Rule #3
        FOR J = 0 TO 14     ' (15 * 1 minute = 0.25 Hour) 

        Water = 1               ' Turn on water soleniod 
        PAUSE 60000 ' 1 minute time delay 
        NEXT J

    ENDIF
    J = 0       ' Clear counter
    

' Note- Rule#4 is alreday built in since none of the above conditions are true
    

    RETURN      ' Return to Main Program

'

'

' Fault Subroutine

' ----------------

' Looks for a WDT Lockup of Microcontroller
Error:

    IF STATUS.4 = 0 THEN    ' Checks for a WDT Lockup in the status register
    Fault = 1       ' Turn on Fault LED
    Water = 1       ' Turn off water, if on
    STOP            ' Stops Program Execution
    ENDIF
    RETURN          ' Return to main program

